Signed Curvature of a Plane Curve

The general notion of curvature of a curve in $\mathbb{R}^n$ does not take account of sign. For curves in $\mathbb{R}^2$ there is a more refined notion of signd that does.

```
κs[α_][t_] := (α''[t] . RotationTransform[π/2][α'[t]]) / (α'[t].α'[t])^(3/2)
```

α must be a 2D vector valued function and t must be a real number.

RotationTransform is one of Mathematica's geometric functions, which rotates a vector through a given angle:

```
a[t_] := {x[t], y[t]}
```

```
κs[a][t]
```

$$\frac{-y'[t]\, x''[t] + x'[t]\, y''[t]}{\left(x'[t]^2 + y'[t]^2\right)^{3/2}}$$

The above definition works for functions $\mathbb{R} \to \mathbb{R}^2$. We give another one for a pair of functions:

```
κs[{x_, y_}][t_] = (-y'[t] x''[t] + x'[t] y''[t]) / (x'[t]^2 + y'[t]^2)^(3/2);
```

and one more for a pair of expressions.

```
κs[{x_, y_}, t_] := κs[{Function[x /. t → #], Function[y /. t → #]}][t]
```

Here we calculate signed curvature of a circle given as a pair of expressions.

```
κs[{Sin[t], Cos[t]}, t] // Simplify
```

-1

Here is the same thing for a pair of functions:

```
κs[{Cos , Sin}][t] // Simplify
```

1

```
κs[{Cos, Sin}][t] // Simplify
```

1

and here for a single function with values in $\mathbb{R}^2$

```
κs[{Cos[#], Sin[#]} &][t] // Simplify
```

1

```
bulletnose[a_, b_][t_] := {a Cos[t], b Cot[t]}
```

$\kappa s[\text{bulletnose}[a, b]][\frac{\pi}{3}]$

$$-\frac{2\,a\,b}{\left(\frac{3\,a^2}{4}+\frac{16\,b^2}{9}\right)^{3/2}}$$

```
Simplify[%]
```

$$-\frac{432\,a\,b}{\left(27\,a^2+64\,b^2\right)^{3/2}}$$

The positive function $\frac{1}{|\kappa_s(\alpha)|}$ is called the radius of curvature of α.

Built-in FrenetSerretSystem function

Mathematica has a built in function that calculates the curvature of a curve in $\mathbb{R}^n$ for any positive integer $n \geq 2$. In two dimensions

```
FrenetSerretSystem[{u[t], v[t]}, t]
```

returns a list consisting of the signed curvature, the unit tangent and unit normal vectors at the point corresponding to t .

```
Simplify[FrenetSerretSystem[bulletnose[a, b][t], t] /. t → Pi/3]
```

$$\left\{\left\{-\frac{432\,a\,b}{\left(27\,a^2+64\,b^2\right)^{3/2}}\right\},\ \left\{\left\{-\frac{a}{\sqrt{a^2+\frac{64\,b^2}{27}}},\ -\frac{8\,b}{\sqrt{27\,a^2+64\,b^2}}\right\},\ \left\{\frac{8\,b}{\sqrt{27\,a^2+64\,b^2}},\ -\frac{a}{\sqrt{a^2+\frac{64\,b^2}{27}}}\right\}\right\}\right\}$$

Evolutes

Circles of curvature

A point $p \in \mathbb{R}^2$ is called the center of curvature at q of a curve $\alpha : (a, b) \to \mathbb{R}^2$ provided there is a circle γ with centre p which is tangent to α at q and such that the curvature of α and γ are the same at q. Since the radius of curvature of a circle is equal to its radius, the radius of the circle of curvature is the radius of curvature.

The centers of the circles of curvature of a curve α form a curve called the evolute of α. The equation of the evolute is:

$$\text{evolute}[f\_][t\_] := \text{Simplify}\left[\frac{\text{RotationTransform}[\frac{\pi}{2}][f'(t)]}{\sqrt{f'(t).f'(t)}\ \kappa_s(f)(t)} + f(t)\right]$$

```
gg[t_] := {x[t], y[t]}

evolute[gg][t]
```

$$\left\{x[t]+\frac{y'[t]\left(x'[t]^2+y'[t]^2\right)}{y'[t]\,x''[t]-x'[t]\,y''[t]},\ y[t]+\frac{x'[t]\left(x'[t]^2+y'[t]^2\right)}{-y'[t]\,x''[t]+x'[t]\,y''[t]}\right\}$$

```
evolute[{x_, y_}][t_] = evolute[gg][t]
```

$$\left\{x[t]+\frac{y'[t]\left(x'[t]^2+y'[t]^2\right)}{y'[t]\,x''[t]-x'[t]\,y''[t]},\ y[t]+\frac{x'[t]\left(x'[t]^2+y'[t]^2\right)}{-y'[t]\,x''[t]+x'[t]\,y''[t]}\right\}$$

As earlier, we define the evolute for curves given by pairs of functions and pairs of expressions.

```
evolute[{x_, y_}, t_] := evolute[{Function[x /. t → #], Function[y /. t → #]}][t]

evolute[{a Sin[#] &, b Cos[#] &}][t] // Simplify
```

$$\left\{\frac{\left(a^2-b^2\right)\mathrm{Sin}[t]^3}{a},\ \frac{\left(-a^2+b^2\right)\mathrm{Cos}[t]^3}{b}\right\}$$

```
evolute[{a Sin[#], b Cos[#]} &][t]
```

$$\left\{\frac{\left(a^2-b^2\right)\mathrm{Sin}[t]^3}{a},\ \frac{\left(-a^2+b^2\right)\mathrm{Cos}[t]^3}{b}\right\}$$

```
gr = ParametricPlot[
  {{2 Cos[t], 3 Sin[t]}, evolute[{2 Cos[#] &, 3 Sin[#] &}][t]}, {t, 0, 2 Pi}]
```



```
Manipulate[Show[gr,
  Graphics[{Red, PointSize[0.02], Point[evolute[{2 Cos[#] &, 3 Sin[#] &}][t]],
    Circle[evolute[{2 Cos[#] &, 3 Sin[#] &}][t],
     1 / Abs[κs[{2 Cos[#] &, 3 Sin[#] &}][t]]]}],
  PlotRange → 8], {t, 0, 2 Pi}, SaveDefinitions → True]
```



Involutes

The involute is the operation of taking the inverse to the operation that assigns to a curve its evolute. The relation between evolute and involute is analogous to the relation between differentiation and integration. The deinition of involute uses integration but this very rarely can be successfully done symbolically. Therefore we will mostly usea numerical definition which is suitable for finding involutes graphically. To define an involute, we need to give not only a curve but also a starting value *c* of the parameter.

```
nInvolute[α_, c_][t_] :=
 - (α′[t] NIntegrate[Norm[α′[s]], {s, c, t}]) / Norm[α′[t]] + α[t]
```

```
nInvolute[{Cos[#], Sin[#]} &, 0][0.3]
```

```
{1.04399, 0.00891926}
```

Making the graphic of an involute even of a cirlce takes a long time.

```
ParametricPlot[nInvolute[{Cos[#], Sin[#]} &, 0][t], {t, 0, 8 Pi}]
```



We also give a symbolic definition of involute. In those cases when involutes can be found explicitely, drawing them becomes much quicker.

$$\text{involute}(\alpha\_, c\_)(t\_) := \text{Assuming}\Big[t \in \mathbb{R}, \text{Simplify}\Big[\Big(-\alpha'(t)\ \text{Integrate}\Big[\sqrt{\alpha'(s).\alpha'(s)}, \{s, c, t\}, \text{Assumptions} \to \{t >= c >= 0\}\Big]\Big) \Big/ \Big(\sqrt{\alpha'(t).\alpha'(t)}\Big) + \alpha(t)\Big]\Big]$$

```
involute[{x_, y_}, c_, t_] := involute[Function[{x, y} /. t → #], c][t]
```

```
inv[t_] = involute[{2 Cos[#], 3 Sin[#]} &, 0][t]
```

$$\Big\{2\,\text{Cos}[t] + \Big(6\,\Big(\text{EllipticE}\Big[\pi\,\text{FractionalPart}\Big[\frac{t}{\pi}\Big], \frac{5}{9}\Big] + 2\,\text{EllipticE}\Big[\frac{5}{9}\Big]\,\text{IntegerPart}\Big[\frac{t}{\pi}\Big]\Big)\,\text{Sin}[t]\Big) \Big/ \Big(\sqrt{9\,\text{Cos}[t]^2 + 4\,\text{Sin}[t]^2}\Big),$$
$$3\,\text{Sin}[t] - \Big(9\,\text{Cos}[t]\,\Big(\text{EllipticE}\Big[\pi\,\text{FractionalPart}\Big[\frac{t}{\pi}\Big], \frac{5}{9}\Big] + 2\,\text{EllipticE}\Big[\frac{5}{9}\Big]\,\text{IntegerPart}\Big[\frac{t}{\pi}\Big]\Big)\Big) \Big/ \Big(\sqrt{9\,\text{Cos}[t]^2 + 4\,\text{Sin}[t]^2}\Big)\Big\}$$

```
ParametricPlot[inv[t], {t, 0, 8 Pi}]
```




```
ParametricPlot[Evaluate[involute[{Cos[#], Sin[#]} &, 0][t]], {t, 0, 8 Pi}]
```



```
g = ParametricPlot[Evaluate[involute[{2 Cos[#], 3 Sin[#]} &, 0][t]], {t, 0, 8 Pi}]
```




```
ParametricPlot[Evaluate[
  {{2 Cos[t], 3 Sin[t]}, involute[{2 Cos[#], 3 Sin[#]} &, 0][t]}], {t, 0, 4 Pi}]
```



```
h = involute[a { Cos[t], Sin[t]}, b, t] // Factor
```

{a (Cos[t] - b Sin[t] + t Sin[t]), a (b Cos[t] - t Cos[t] + Sin[t])}

```
g = evolute[h, t] // FullSimplify
```

{a Cos[t], a Sin[t]}

```
FullSimplify[involute[g, 0, t], Assumptions → t > 0]
```

{a (Cos[t] + t Sin[t]), a (-t Cos[t] + Sin[t])}